
    1                 /       CPU IDENTIFICATION PROGRAM.
    2   
    3                 /       LAST EDIT: 03-OCT-2016 CJL
    4   
    5                 /       MUST BE ASSEMBLED WITH THE ’/J’ COMMAND-LINE SWITCH OFF IN P?S/8 PAL OR THE
    6                 /       ’/F’ COMMAND-LINE SWITCH OFF IN OS/8 PAL8.
    7   
    8                 /       THIS PROGRAM IDENTIFIES THE PARTICULAR MODEL OF PDP-8 THE PROGRAM IS RUNNING
    9                 /       ON.  THIS IS ACCOMPLISHED USING A SERIES OF TESTS FOR QUIRKS THAT APPLY TO THE
   10                 /       VARIOUS MODELS.
   11   
   12                 /       NOTE: MUCH OF THE CODE IN THIS PROGRAM WAS OBTAINED FROM THE KERMIT-12 SOURCE
   13                 /       FILES; SOME MODIFICATIONS HAVE BEEN MADE THAT WILL EVENTUALLY BE APPLIED TO
   14                 /       KERMIT-12 TO BETTER IDENTIFY THE COMPUTER.  NOTE: KERMIT-12 USES A
   15                 /       MODEL-DEPENDENT COMMAND PROMPT TO HELP ASSIST IN DETERMINING WHICH KERMIT IS
   16                 /       CURRENTLY BEING ADDRESSED; IN CERTAIN CIRCUMSTANCES, TWO DIFFERENT SYSTEMS ARE
   17                 /       IN USE.  IT CAN SOMETIMES BE CONFUSING WHICH COMPUTER IS CURRENTLY IN EFFECT,
   18                 /       THUS, USING MODEL-SPECIFIC PROMPTS CAN AID IN PREVENTING CONFUSION.  AS
   19                 /       CURRENTLY IMPLEMENTED AS OF THIS WRITING, KERMIT-12 CANNOT QUITE DISTINGUISH
   20                 /       BETWEEN DECMATE SYSTEMS AND OTHER SYSTEMS BASED ON THE 6120 CHIP SUCH AS THE
   21                 /       CPU-8 OR GIZMO.  WHILE THIS ASPECT OF THE PROBLEM IS PURELY COSMETIC,
   22                 /       KERMIT-12 HAS CONFIGURATION ISSUES WHEN RUN ON THESE PARTICULAR SYSTEMS.  AS
   23                 /       SUCH, FUTURE RELEASES OF KERMIT-12 WILL NOT ONLY MODIFY THE COMMAND PROMPT,
   24                 /       BUT THE INTERNAL LOGIC WILL PROPERLY SELF-CONFIGURE USING ANALOGOUS CODING TO
   25                 /       THAT WITHIN THIS CPU IDENTIFICATION PROGRAM.
   26   
   27                 /       AN OLDER PROGRAM SEGMENT ATTEMPTING CPU IDENTIFICATION IS CONTAINED WITHIN THE
   28                 /       ORIGINAL RELEASE OF FOCAL, 1969; HOWEVER, THAT [OPTIONAL OVERLAY] HAS SEVERAL
   29                 /       KNOWN PROBLEMS:
   30   
   31                 /       1) NEWER MODELS [INCLUDING DECMATES] COULD NOT BE FORESEEN; AS SUCH, THESE
   32                 /          SYSTEMS ARE MISIDENTIFIED.
   33   
   34                 /       2) THE CODING USED TO IDENTIFY THE PDP-8/I AND PDP-8/L IS FLAWED; CERTAIN
   35                 /          CONFIGURATIONS OF PDP-8/I WILL BE MISIDENTIFIED AS PDP-8/L BECAUSE THE
   36                 /          METHOD CHOSEN IS TO USE SOFTWARE TIMING BASED ON THE CONSOLE OUTPUT
   37                 /          AS A PRIMITIVE CLOCK.  THERE ARE MULTIPLE REASONS WHY THIS METHOD CANNOT
   38                 /          BE TRUSTED RANGING FROM MEMORY TIMING PROBLEMS OF THE CPU TO CONSOLE DEVICE
   39                 /          UPGRADES THAT RUN FASTER THAN THE TELETYPE MODELS 33 OR 35.  THE PROGRAM’S
   40                 /          DEPENDENCE ON NOMINAL TIMING VALUES THAT MIGHT NOT APPLY WOULD TEND TO
   41                 /          INCREASE THE NUMBER OF MACHINES MISIDENTIFIED EVEN DURING THE ECONOMIC
   42                 /          LIFETIME OF THE PDP-8/I [AND BEYOND].
   43   
   44                 /       THE PROGRAM IMPLEMENTED HERE USES AN ALTERNATE METHOD THAT ALWAYS WORKS.  THE
   45                 /       PDP-8/L INHIBITS ALL TYPE III OPERATE INSTRUCTIONS WHILE THE PDP-8/I ALWAYS
   46                 /       ALLOWS CERTAIN INSTRUCTIONS WITHIN THIS OPERATE GROUP, EVEN IF EAE IS NOT
   47                 /       PRESENT.  THE ACCUMULATOR CAN BE CLEARED USING THE AC CLEAR OPERATION IN GROUP
   48                 /       III ONLY ON THE PDP-8/I; THE ACCUMULATOR WILL NOT CLEAR ON THE PDP-8/L.

   50                 /       LINC-8 INSTRUCTION DEFINITIONS.
   51   
   52           6171          IAAC=   6171                    /LOAD LINC "A" REGISTER INTO ACCUMULATOR.
   53           6167          IACA=   6167                    /LOAD ACCUMULATOR INTO LINC "A" REGISTER.
   54   
   55                 /       PDP-12 INSTRUCTION DEFINITIONS.
   56   
   57           0017          COM=    0017                    /COMPLEMENT ACCUMULATOR [IN LINC MODE].
   58           6141          LINC=   6141                    /SWITCH FROM PDP-8 MODE TO LINC MODE ON PDP-12.
   59           0002          PDP=    0002                    /SWITCH FROM LINC MODE TO PDP-8 MODE ON PDP-12.
   60   
   61                 /       NUMERIC LOAD DEFINITIONS.  [NOTE: ONLY USE THOSE UNIVERSAL TO ALL MODELS!]
   62   
   63           7240          NL7777= CLA CMA                 /LOAD ACCUMULATOR WITH 7777.
   64   
   65                 /       PROGRAM DEFINITIONS.
   66   
   67           7600          SBOOT=  7600                    /EXIT LOCATION FOR MOST OPERATING SYSTEMS.
   68   
   69           0000          FIELD   0                       /PROGRAM SHOULD BE LOADED INTO FIELD 0.
   70   
   71          *0002          *PDP                            /OVER POTENTIALLY RELEVANT MEMORY LOCATION.
   72   
   73   000002  0000          ZBLOCK  1                       /THIS IS NEEDED FOR CERTAIN PDP-8/I SYSTEMS.
   74   
   75          *0010          *10                             /GET TO AUTO-INDEX AREA.
   76   
   77   000010  0000  XR0,    .-.                             /AUTO-INDEX REGISTER 0.

   79                 /       THE PROGRAM STARTS HERE.
   80   
   81          *0200          *200                            /THE USUAL STARTING LOCATION.
   82   
   83   000200  7300  START,  CLA CLL                         /CLEAN UP.
   84   
   85                 /       FOR MOST PROGRAMS, THE STATE OF THE CONSOLE OUTPUT FLAG CAN BE ASSUMED WHERE
   86                 /       THE WORST CASE PROBLEM IS TO "SPLAT" THE PREVIOUS CHARACTER IN THE PROCESS OF
   87                 /       BEING PRINTED; THE OUTPUT FLAG WILL RAISE REGARDLESS OF THIS POTENTIAL
   88                 /       COSMETIC PROBLEM.  HOWEVER, PART OF THE LOGIC OF THIS PROGRAM INCLUDES THE
   89                 /       ABILITY TO PROPERLY IDENTIFY DECMATE SYSTEMS [AS OPPOSED TO OTHER SYSTEMS ALSO
   90                 /       BASED ON THE 612 CHIP].  AS SUCH, WE HAVE TO HANDLE ALL POTENTIAL SITUATIONS.
   91   
   92                 /       1) THE PROGRAM IS RUNNING STANDALONE AND WAS STARTED FROM THE FRONT PANEL.  IN
   93                 /          THIS SITUATION, THE CONSOLE OUTPUT FLAG IS NOT SET NOR WILL IT COME UP IN A
   94                 /          REASONABLE AMOUNT OF TIME; ONCE A REASONABLE TIMEOUT HAS OCCURRED, THE
   95                 /          PROGRAM SHOULD OUTPUT AN OPENING CHARACTER.  FOR THIS PURPOSE, THE <NUL>
   96                 /          CHARACTER IS PREFERRED SINCE ON MOST SYSTEMS IT HAS NO SIGNIFICANCE OTHER
   97                 /          THAN A ONE-CHARACTER DELAY; ON TELETYPE MODEL 33 AND 35 BASED SYSTEMS, THE
   98                 /          OUTPUT WILL CAUSE AN ASSOCIATED SOUND; HOWEVER, THE CARRIAGE WILL NOT BE
   99                 /          ADVANCED THUS THIS METHOD IS TOTALLY INNOCUOUS.
  100   
  101                 /       2) THE PROGRAM COULD BE RUNNING FROM AN OPERATING SYSTEM SUCH AS OS/8 WHERE
  102                 /          THERE IS NO ABSOLUTE RULE REGARDING THE CONSOLE FLAG CONVENTION.  BY AN
  103                 /          INACCURATE CONSENSUS, MANY PROGRAMS MERELY ASSUME THERE IS NO CONSOLE
  104                 /          OUTPUT IN PROGRESS; CHARACTERS ARE OUTPUT AND THEN THE FLAG IS WAITED FOR.
  105                 /          WHILE THIS MIGHT WORK MOST OF THE TIME, IT IS NOT FOOLPROOF AND ALSO USES
  106                 /          THE INEFFICIENT METHOD THAT DOES NOT OVERLAP PROCESSING WITH CONSOLE
  107                 /          OUTPUT.  AS SUCH, THIS CASE SHOULD BE HANDLED MUCH LIKE 1) ABOVE WITH THE
  108                 /          PROVISO THAT PROPER TESTING AFTER THE TIMEOUT MUST BE PERFORMED.
  109   
  110                 /       3) THE PROGRAM MAY BE RUNNING UNDER P?S/8 WITHOUT THE LOGICAL CONSOLE OVERLAY
  111                 /          ENABLED.  IN THIS CASE, THE PROGRAM HAS A RIGHT TO ASSUME A CHARACTER IS
  112                 /          EITHER IN THE PROCESS OF PRINTING OR THE DONE FLAG HAS ALREADY RAISED.  IF
  113                 /          THIS MODE OF OPERATION WERE GUARANTEED TO APPLY, THE PROGRAM COULD BE
  114                 /          WRITTEN TO THE ORIGINAL P?S/8 SPECIFICATION WHERE OVERLAPPING PROCESSING
  115                 /          AND CONSOLE OUTPUT CAN BE DONE WITH IMPUNITY WITHOUT ANY INITIAL OUTPUT
  116                 /          CONSIDERATION.  UNFORTUNATELY, P?S/8 NOW SUPPORTS THE LOGICAL CONSOLE
  117                 /          OVERLAY WHICH CHANGES THIS CONVENTION SIGNIFICANTLY. [SEE 4) BELOW.]
  118   
  119                 /       4) THE PROGRAM MAY BE RUNNING UNDER P?S/8 WITH THE LOGICAL CONSOLE OVERLAY
  120                 /          ENABLED.  OUTPUT DIRECTED TO THE LOGICAL CONSOLE DEVICE REQUIRES THE
  121                 /          PROGRAM TO AVOID ANY PHYSICAL CONSOLE OUTPUT AND INSTEAD DIRECT ALL OUTPUT
  122                 /          TO LOGICAL SUBROUTINE CALLS IN EXTENDED MEMORY.  AS SUCH, ALL SUCH USE OF
  123                 /          THE CONSOLE DEVICE IS DONE WITH THE PROVISO THAT THE PHYSICAL CONSOLE
  124                 /          DEVICE MAY BE UNDER THE CONTROL OF THE LOGICAL CONSOLE OVERLAY WHICH MAY OR
  125                 /          MAY NOT BE DIRECTED TO THE DEVICE 03 NORMAL PHYSICAL CONSOLE.  HOWEVER, THE
  126                 /          LOGIC OF THIS PROGRAM REQUIRES PHYSICAL ACCESS TO THE CONSOLE DEVICE; AS
  127                 /          SUCH, IT IS RECOMMENDED THE PROGRAM EXIT [VIA BRANCH TO 07600] TO ALLOW THE
  128                 /          POTENTIALLY ENABLED LOGICAL CONSOLE OVERLAY TO PROPERLY BE RESET.  THE
  129                 /          PROGRAM WILL BE RUN USING A PROPER TIMEOUT SUCH AS IS DESCRIBED IN 1) ABOVE
  130                 /          TO ENSURE THE CONSOLE OUTPUT IS NOT GARBLED.  NOTE: A MORE COMPLEX PROGRAM
  131                 /          COULD BE CONSTRUCTED TO BE FULLY CONFORMING TO P?S/8 STANDARDS FOR THE
  132                 /          CONSOLE DEVICE, BUT THIS IS BEYOND THE SCOPE OF THIS PROGRAM AS WRITTEN.

  134                 /       THE BEST WAY TO HANDLE ALL OF THE ABOVE CONSIDERATIONS CONSISTS OF THE
  135                 /       FOLLOWING:
  136   
  137                 /       1) PERFORM A REASONABLE TIMEOUT THAT INCLUDES CHECKING THE CONSOLE OUTPUT FLAG
  138                 /          DURING THE TIMEOUT LOOP.
  139   
  140                 /       2) IF THE FLAG RAISES THUS ENDING THE TIMEOUT INTERVAL, FURTHER OUTPUT CAN BE
  141                 /          PERFORMED WITH IMPUNITY.  SINCE DECMATE DETECTION HASN’T BEEN PERFORMED
  142                 /          YET, OUTPUT A <NUL> CHARACTER WITHOUT WAITING FOR THE DONE FLAG.
  143   
  144                 /       3) IF THE FLAG DID NOT RAISE DURING THE TIMEOUT INTERVAL, THE OUTPUT MUST BE
  145                 /          INITIALIZED WITH THE <NUL> CHARACTER OUTPUT AFTERWARDS.  ALL PRINTING IN
  146                 /          THE REST OF THE PROGRAM CAN USE THE METHOD OF OVERLAPPING PROCESSING AND
  147                 /          OUTPUT WAIT FOR MAXIMUM EFFICIENCY WITHOUT REGARD FOR WHETHER 3) OR 2)
  148                 /          ABOVE APPLIES.
  149   
  150                 /       4) IF IT IS DETERMINED THE CPU IS THE 6120 CHIP, A NORMAL WAIT FOR THE CONSOLE
  151                 /          OUTPUT DONE FLAG WILL BE PERFORMED.  ONCE THE OUTPUT FLAG SKIPS, THE
  152                 /          "TSF" INSTRUCTION WILL BE PERFORMED AGAIN.  IF THE INSTRUCTION CAUSES A
  153                 /          SKIP, THE MACHINE IS NOT A DECMATE.  DETECTION OF A DECMATE WILL CHANGE THE
  154                 /          EVENTUAL DETECTION DISPLAY CONTENTS; ANOTHER <NUL> CHARACTER WILL BE OUTPUT
  155                 /          TO SETUP THE PROPER OVERLAP USED IN ALL MESSAGE PRINTING ROUTINES.
  156   
  157                 /       SETUP THE OUTER TIMEOUT LOOP COUNTER.  THIS CAN BE ADJUSTED FOR UNUSUAL
  158                 /       SITUATIONS SUCH AS "HOME-BREW" MACHINES THAT ARE VERY FAST COMPARED TO
  159                 /       STANDARD MODELS YET HAVE LOW BAUD RATE SERIAL CONSOLE INTERFACES.
  160   
  161   000201  1377          TAD     (-30)                   /SETUP A REASONABLE
  162   000202  3305          DCA     OUTER                   /WAIT TIME.
  163   000203  6041  WATLUP, TSF                             /DID THE OUTPUT FLAG COME UP?
  164   000204  7410          SKP                             /SKIP IF NOT.
  165   000205  5212          JMP     FLAGUP                  /JUMP IF IT DID COME UP.
  166   000206  2303          ISZ     INNER                   /WAIT FOR A
  167   000207  5206          JMP     .-1                     /BIT OF TIME.
  168   000210  2305          ISZ     OUTER                   /WAITED LONG ENOUGH?
  169   000211  5203          JMP     WATLUP                  /NO, KEEP GOING.
  170   
  171                 /       COMES HERE BECAUSE THE FLAG DID COME UP OR WE WAITED LONG ENOUGH AND THE FLAG
  172                 /       NEVER RAISED.  WE CAN OUTPUT A <NUL> CHARACTER TO INITIALIZE ALL FURTHER
  173                 /       PRINTING OPERATIONS.
  174   
  175   000212  6046  FLAGUP, TLS                             /OUTPUT A <NUL> CHARACTER NOW.
  176   
  177                 /       NOW FIND OUT WHAT KIND OF MACHINE THIS IS.
  178   
  179   000213  4776          JMS I   (MACHINE)               /CALL IDENTIFICATION ROUTINE.
  180   000214  3304          DCA     MODEL                   /STORE IT.
  181   000215  1304          TAD     MODEL                   /GET IT BACK.
  182   000216  1375          TAD     (-13)                   /COMPARE TO 6120 VALUE.
  183   000217  7640          SZA CLA                         /SKIP IF IT MATCHES.
  184   000220  5230          JMP     PRCPU                   /JUMP IF ANY OTHER MODEL.

  186                 /       THE CPU IS A 6120; IS THIS A DECMATE?
  187   
  188   000221  6041          TSF                             /OUTPUT FLAG UP?
  189   000222  5221          JMP     .-1                     /NOT YET, WAIT FOR IT.
  190   
  191                 /       IF THIS IS A DECMATE, THE FLAG WAS CLEARED BY THE "TSF" INSTRUCTION.
  192   
  193   000223  6041          TSF                             /FLAG STILL UP?
  194   000224  7410          SKP                             /SKIP IF NOT.
  195   000225  5230          JMP     PRCPU                   /NOT A DECMATE; JUST HANDLE NORMALLY.
  196   
  197                 /       THIS MACHINE IS A DECMATE; INDICATE THIS FOR THE OUTPUT DISPLAY AND OUTPUT AN
  198                 /       ADDITIONAL <NUL> CHARACTER RIGHT NOW SO FURTHER PRINTING WILL FUNCTION.
  199   
  200   000226  2304          ISZ     MODEL                   /SET MODEL TO 14: 6120 ON DECMATE.
  201   000227  6046          TLS                             /OUTPUT ANOTHER <NUL> CHARACTER.
  202   
  203                 /       COMES HERE WITH THE OUTPUT FLAG IN PROGRESS FOR PRINTING THE NOW FULLY
  204                 /       DETERMINED MACHINE MODEL TYPE.
  205   
  206   000230  4275  PRCPU,  JMS     CRLF                    /PRINT <CR>/<LF> FIRST.
  207   000231  1374          TAD     (OPNMSG-1)              /GET POINTER TO THE OPENING MESSAGE.
  208   000232  4242          JMS     PSTRING                 /PRINT IT.
  209   000233  1373          TAD     (MDLPTRS)               /GET TABLE OF MODELS POINTER.
  210   000234  1304          TAD     MODEL                   /INDEX TO THE RELEVANT ONE.
  211   000235  3306          DCA     TEMP                    /STASH THE POINTER.
  212   000236  1706          TAD I   TEMP                    /GET THE PROPER STRING MESSAGE POINTER
  213   000237  4242          JMS     PSTRING                 /PRINT THE MODEL-SPECIFIC STRING.
  214   000240  4275          JMS     CRLF                    /PRINT <CR>/<LF> BEFORE EXIT.
  215   000241  5772          JMP I   (7600)                  /EXIT TO O/S. HLT [7402] IF STANDALONE.
  216   
  217                 /       STRING PRINTING ROUTINE.
  218   
  219                 /       CALL WITH THE ACCUMULATOR POINTING TO THE TEXT STRING [-1].
  220   
  221   000242  0000  PSTRING,.-.                             /STRING PRINT ROUTINE.
  222   000243  3010          DCA     XR0                     /STASH THE POINTER.
  223   000244  1410  PLOOP,  TAD I   XR0                     /GET A PAIR OF CHARACTERS TO PRINT.
  224   000245  3306          DCA     TEMP                    /SAVE THEM.
  225   000246  1306          TAD     TEMP                    /GET THEM BACK.
  226   000247  7012          RTR;RTR;RTR                     /MOVE HIGH-ORDER TO LOW-ORDER AC BITS.
  227   000250  7012 
  228   000251  7012 
  229   000252  4256          JMS     P6CH                    /PRINT ORIGINAL HIGH-ORDER.
  230   000253  1306          TAD     TEMP                    /GET THEM BACK AGAIN.
  231   000254  4256          JMS     P6CH                    /PRINT THE LOW-ORDER NOW.
  232   000255  5244          JMP     PLOOP                   /GO DO NEXT PAIR.

  234   000256  0000  P6CH,   .-.                             /PRINT A SIXBIT TEXT CHARACTER ROUTINE.
  235   000257  0371          AND     (77)                    /JUST THE LATEST CHARACTER BITS.
  236   000260  7450          SNA                             /SKIP IF NOT AT END OF MESSAGE.
  237   000261  5642          JMP I   PSTRING                 /RETURN TO ORIGINAL CALLER.
  238   000262  1370          TAD     (40)                    /INVERT QUADRANT FOR ASCII CONVERSION.
  239   000263  0371          AND     (77)                    /JUST CHARACTER BITS.
  240   000264  1370          TAD     (40)                    /NOW CONVERTED BACK TO ASCII.
  241   000265  4267          JMS     P7CH                    /PRINT IT.
  242   000266  5656          JMP I   P6CH                    /RETURN TO CALLER.
  243   
  244   000267  0000  P7CH,   .-.                             /SEVEN-BIT ASCII PRINT ROUTINE.
  245   000270  6041          TSF                             /FLAG UP?
  246   000271  5270          JMP     .-1                     /NO, WAIT FOR IT.
  247   000272  6046          TLS                             /YES, PRINT THE CHARACTER.
  248   000273  7200          CLA                             /CLEAN UP
  249   000274  5667          JMP I   P7CH                    /RETURN TO CALLER.
  250   
  251   000275  0000  CRLF,   .-.                             /<CR>, <LF> PRINTING ROUTINE.
  252   000276  1367          TAD     ("M&37)                 /GET <CR> CODE.
  253   000277  4267          JMS     P7CH                    /PRINT IT.
  254   000300  1366          TAD     ("J&37)                 /GET <LF> CODE.
  255   000301  4267          JMS     P7CH                    /PRINT IT.
  256   000302  5675          JMP I   CRLF                    /RETURN TO CALLER.
  257   
  258   000303  0000  INNER,  .-.                             /INNER TIMEOUT COUNTER.
  259   000304  0000  MODEL,  .-.                             /MACHINE MODEL TEMPORARY.
  260   000305  0000  OUTER,  .-.                             /OUTER TIMEOUT COUNTER.
  261   000306  0000  TEMP,   .-.                             /MULTI-PURPOSE TEMPORARY.
  262   
  263   000307  0616  MDLPTRS,UNKNMSG-1                       /0=UNKNOWN MODEL.
  264   000310  0627          PDP5MSG-1                       /1=PDP-5.
  265   000311  0633          PDP8MSG-1                       /2=PDP-8.
  266   000312  0637          PDP8SMSG-1                      /3=PDP-8/S.
  267   000313  0644          LINC8MSG-1                      /4=LINC-8.
  268   000314  0651          PDP8IMSG-1                      /5=PDP-8/I.
  269   000315  0656          PDP8LMSG-1                      /6=PDP-8/L.
  270   000316  0663          PDP12MSG-1                      /7=PDP-12.
  271   000317  0670          KK8FMSG-1                       /10=KK8F [PDP-8/E, F, M, 8A-600 SERIES].
  272   000320  0706          KK8AMSG-1                       /11=KK8A [PDP-8/A 400 OR SIMILAR].
  273   000321  0713          CHP6100-1                       /12=6100 CHIP [VT78 OR INTERCEPT I].
  274   000322  0730          CHP6120-1                       /13=6120 CHIP [CPU8 OR GIZMO].
  275   000323  0745          DMATMSG-1                       /14=DECMATE.
  276   
  277          *0366 
        000366  0012 
        000367  0015 
        000370  0040 
        000371  0077 
        000372  7600 
        000373  0307 
        000374  0577 
        000375  7765 
        000376  0400 
        000377  7750 
               *0400          PAGE

  279                 /       CPU IDENTIFICATION ROUTINE.
  280   
  281                 /       CALL WITH:
  282   
  283                 /       AC, LINK UNIMPORTANT
  284   
  285                 /       JMS I   (MACHINE)               /CALL THE ROUTINE.
  286   
  287                 /       UPON EXIT, THE ACCUMULATOR WILL CONTAIN THE IDENTIFIED MODEL:
  288   
  289                 /       AC CONTENTS     CPU TYPE
  290   
  291                 /            0          UNKNOWN MODEL [NON-DEC]
  292                 /            1          PDP-5
  293                 /            2          ORIGINAL PDP-8
  294                 /            3          PDP-8/S
  295                 /            4          LINC-8
  296                 /            5          PDP-8/I
  297                 /            6          PDP-8/L
  298                 /            7          PDP-12
  299                 /           10          KK8F CPU [PDP-8E, F, M OR PDP-8/A 600 SERIES]
  300                 /           11          KK8A CPU [PDP-8/A 400 OR SIMILAR]
  301                 /           12          6100 CHIP [VT78 OR INTERSIL INTERCEPT I]
  302                 /           13          6120 CHIP [DECMATE, GIZMO OR CPU-8]
  303   
  304   
  305   000400  0000  MACHINE,.-.                             /MACHINE IDENTIFYING ROUTINE.
  306   000401  7300          CLA CLL                         /CLEAN UP.
  307   000402  1000          TAD     0000                    /GET LOCATION 0000 [PC ON PDP-5].
  308   000403  3266          DCA     MCHTEMP                 /SAVE IT [WILL LATER RESTORE ON MOST MODELS].
  309   
  310                 /       FIRST TEST: PDP-5.
  311   
  312                 /       ON THE PDP-5 LOCATION 0000 IN MEMORY IS USES AS THE PROGRAM COUNTER, AN INEPT
  313                 /       ATTEMPT AT "ECONOMY" WHICH CAUSES THE MACHINE TO BE RELATIVELY SLOW DESPITE
  314                 /       THE USE OF HIGH-SPEED CORE MEMORY [QUITE EXPENSIVE AT THE TIME] BECAUSE OF THE
  315                 /       EXTRA CYCLES USED FOR EVERY INSTRUCTION TO MAINTAIN AND UPDATE IT.  IF STORED
  316                 /       INTO, THIS IS EQUIVALENT TO PERFORMING A "JMP" INSTRUCTION IN THE ORDINARY
  317                 /       SENSE; THE VALUE WILL BE INCREMENTED ONE [WHICH IS THE NORMAL OPERATION OF A
  318                 /       PROGRAM COUNTER UNLESS THE INSTRUCTION EXECUTION ALSO SKIPPED].
  319   
  320   000404  1377          TAD     (PDP5-1)                /GET PDP-5 DETECT ADDRESS [WILL BE INCREMENTED].
  321   000405  3000          DCA     0000                    /STORE OVER PROGRAM COUNTER [IF PDP-5].
  322   
  323                 /       EXECUTION CONTINUES HERE FOR ALL MODELS EXCEPT THE PDP-5; ON THE PDP-5 THE
  324                 /       NEXT INSTRUCTION IS PART OF THE ROUTINE EXIT SETTING THE VALUE FOR THE
  325                 /       PDP-5 IN THE ACCUMULATOR.
  326   
  327                 /       RESTORE THE PREVIOUS CONTENTS OF LOCATION 0000 FOR ALL OTHER MODELS.
  328   
  329   000406  1266          TAD     MCHTEMP                 /GET THE PREVIOUS CONTENTS OF LOCATION 0000.
  330   000407  3000          DCA     0000                    /RESTORE IT.

  332                 /       SECOND TEST: PDP-8/I, PDP-8/L, PDP-12.
  333   
  334                 /       THE PDP-8/I "FAMILY" CAN PERFORM A COMBINED ROTATE OPERATION THAT CAN CLEAR
  335                 /       BOTH THE LOW-ORDER AND HIGH-ORDER BIT IN THE SAME INSTRUCTION.  THIS WAS
  336                 /       REMOVED IN NEWER VALUES AND THE COMBINATION WAS REPURPOSED TO ALLOW OTHER
  337                 /       OPERATIONS.  NEITHER THE ORIGINAL PDP-8 NOR LINC-8 ALLOWS COMBINING INCREMENT
  338                 /       AND ROTATE [SINGLE OR DOUBLE] IN THE SAME INSTRUCTION.  NOTE: THIS COULD BE A
  339                 /       USEFUL COMBINATION OF OPERATIONS TO LOAD THE VALUE 3776 INTO THE AC IN ONE
  340                 /       INSTRUCTION IF IT IS CERTAIN EXECUTION IS LIMITED TO ONE OF THESE THREE
  341                 /       MODELS; SUCH USAGE IS QUITE RARE OUTSIDE OF MODEL DETECTION ROUTINES.
  342   
  343   
  344   000410  7354          CLA CLL CMA RAL RAR             /CLEAR THE AC AND LINK, THEN COMPLEMENT THE AC,
  345                                                         /THEN SET ROTATE ONCE IN BOTH DIRECTIONS.
  346   000411  1376          TAD     (-3776)                 /COMPARE TO EXPECTED VALUE.
  347   000412  7650          SNA CLA                         /SKIP IF NOT THESE THREE MODELS.
  348   000413  5234          JMP     PDP8IL                  /SORT OUT WHICH ONE THERE.
  349   
  350                 /       THIRD TEST: PDP-8/S.
  351   
  352                 /       THE PDP-8/S IS NOT A MEMBER OF "THE FAMILY OF 8" BECAUSE IT LACKS CERTAIN
  353                 /       COMBINED OPERATE FUNCTIONS WHICH ARE SUPPORTED BY THE PDP-8 WHICH WAS RELEASED
  354                 /       EARLIER; THIS IS TRULY A "STEP BACKWARDS" THAT IS NOT TOLERATED BY MOST PDP-8
  355                 /       SOFTWARE INCLUDING P?S/8, THE R-L MONITOR SYSTEM, OS/8, COS-300/310, POLY
  356                 /       BASIC [AND ALL OF ITS EDU VARIANTS], 4K/8K/LAB-8/E BASIC, TSS8 ETC.  THE ONLY
  357                 /       SOFTWARE KNOWN TO RUN ON A PDP-8/S IS FOCAL, 1969 [WHICH SELF-MODIFIES TO
  358                 /       CHANGE SOFTWARE TIMING LOOPS] AND A SPECIAL CONFIGURATION OF THE DISK MONITOR
  359                 /       SYSTEM USING A DF32 WITH INCREASED DATA INTERLEAVING TO PREVENT DATA BREAK
  360                 /       OVERRUN.  THE LATTER POINT IS RELATED TO THE LACK OF SUFFICIENT SPEED OF
  361                 /       OPERATION; ALL OTHER RESTRICTIONS ARE A DIRECT RESULT OF NOT SUPPORTING THE
  362                 /       COMBINING COMPLEMENT AND ROTATE OPERATIONS.  THE COMMONLY USED COMBINED
  363                 /       OPERATE INSTRUCTION NL7776 [CLA CLL CMA RAL] IS USED TO DETECT THE PDP-8/S
  364                 /       SINCE AT THIS POINT, IT IS THE ONLY MODEL NOT ALREADY DETECTED THAT CANNOT
  365                 /       PERFORM THIS INSTRUCTION.  [THE PDP-5 ALSO HAS SIMILAR RESTRICTIONS.]
  366   
  367   000414  7344          CLA CLL CMA RAL                 /ATTEMPT TO LOAD THE AC WITH 7776.
  368   000415  1375          TAD     (-7776)                 /COMPARE TO EXPECTED VALUE.
  369   000416  7640          SZA CLA                         /SKIP IF NOT PDP-8/S.
  370   000417  5277          JMP     PDP8S                   /EXIT SETTING AC VALUE FOR THE PDP-8/S.
  371   
  372                 /       FOURTH TEST: PDP-8, LINC-8.
  373   
  374                 /       AT THIS POINT, THE CPU IS EITHER THE ORIGINAL PDP-8 [ALSO USED IN THE LINC-8]
  375                 /       OR A MODEL NEWER THAN THE PDP-12.
  376   
  377                 /       ALL NEWER MODELS CAN PERFORM A BYTE-SWAP [BSW] OPERATION; TESTING FOR THIS
  378                 /       WILL FAIL ON THE ORIGINAL MODEL[S].
  379   
  380   000420  7203          CLA IAC BSW                     /ON NEWER MODELS THIS LOADS THE AC WITH 0100.
  381   000421  1374          TAD     (-100)                  /COMPARE WITH EXPECTED VALUE.
  382   000422  7650          SNA CLA                         /SKIP IF PDP-8 [OR LINC-8].
  383   000423  5247          JMP     PDPOMNIBUS              /JUMP IF NEWER MODEL.



1 1
2 2
3 3
4 4
5 5
6 6
7 7
8 8
9 9

10 10
11 11
12 12
13 13
14 14
15 15
16 16
17 17
18 18
19 19
20 20
21 21
22 22
23 23
24 24
25 25
26 26
27 27
28 28
29 29
30 30
31 31
32 32
33 33
34 34
35 35
36 36
37 37
38 38
39 39
40 40
41 41
42 42
43 43
44 44
45 45
46 46
47 47
48 48
49 49
50 50
51 51
52 52
53 53
54 54
55 55
56 56
57 57
58 58

/     CPU IDENTIFICATION PROGRAM.     P?S PAL V08S    SAT  11-MAR-17  PAGE 9

  385                 /       FIFTH TEST: LINC-8.
  386   
  387                 /       THE LINC-8 HAS SPECIAL INTERFACE REGISTER INSTRUCTIONS USED FOR VARIOUS
  388                 /       CONTROL FUNCTIONS OF THE LINC CPU WHEN IT IS NOT RUNNING.  [WHEN THE LINC
  389                 /       CPU IS RUNNING, NO PDP-8 INSTRUCTIONS ARE EXECUTED.]  IF THESE INSTRUCTIONS
  390                 /       ARE NOT IMPLEMENTED, THE CPU IS THE ORIGINAL PDP-8 MODEL.
  391   
  392   000424  7240          NL7777                          /SETUP NON-ZERO TEST VALUE.
  393   000425  6167          IACA                            /LOAD THE LINC "A" REGISTER.
  394   000426  7200          CLA                             /CLEAN UP.
  395   000427  6171          IAAC                            /READ THE LINC "A" REGISTER.
  396   000430  7001          IAC                             /ADD ONE TO THE RETURNED VALUE [IF LINC-8].
  397   000431  7650          SNA CLA                         /SKIP IF NOT LINC-8.
  398   000432  5276          JMP     LINC8                   /RETURN TO CALLER WITH LINC-8 VALUE IN AC.
  399   
  400                 /       IN FOCAL, 1969, AN ADDITIONAL TEST WAS PERFORMED TO DETERMINE IF THE AX08
  401                 /       LABORATORY PERIPHERAL WAS INSTALLED; IF THE AX08 WAS PRESENT, THIS WAS USED TO
  402                 /       DECLARE THE MACHINE A "LAB-8".  THE LOGICAL PROBLEM WITH THIS IS TWO-FOLD:
  403                 /       1) NEWER LABORATORY PERIPHERALS WERE RELEASED WHICH UNDERCUTS THE NOTION
  404                 /       ENTIRELY, 2) THE AX08 WAS OFTEN HOOKED TO OTHER MODELS SUCH AS THE PDP-8/I,
  405                 /       YET FOCAL, 1969 COULD NEVER DETECT SUCH COMBINATIONS.  SINCE THIS IS NOT A
  406                 /       VALID CPU DETECTION CONSIDERATION, THE PRESENCE OF AN AX08 IS NOT TESTED FOR.
  407   
  408   000433  5300          JMP     PDP8                    /EXIT WITH THE AC SET FOR PDP-8 DETECTED.
  409   
  410                 /       COMES HERE TO DETERMINE IF THE CPU IS A PDP-8/L OR A PDP-8/I [WHICH COULD BE A
  411                 /       PDP-12].  THIS CODE USES THE GROUP III OPERATE DETECTION METHOD WHICH ALWAYS
  412                 /       WORKS.  AS SUCH, THE TIMING PROBLEMS OF THE FOCAL, 1969 TEST ARE AVOIDED.
  413   
  414   000434  7240  PDP8IL, NL7777                          /SETUP NON-ZERO AC FOR TEST.
  415   000435  7601          CLA!401                         /ATTEMPT GROUP III AC CLEAR OPERATION.
  416   000436  7640          SZA CLA                         /SKIP IF THE AC DID CLEAR.
  417   000437  5274          JMP     PDP8L                   /RETURN WITH AC SET FOR PDP-8/L DETECTION.
  418   
  419                 /       THE CPU IS A PDP-8/I.  CHECK TO SEE IF THE MACHINE IS A PDP-12 [ORIGINALLY
  420                 /       KNOWN AS THE LINC-8/I].
  421   
  422   000440  6141          LINC                            /GO INTO LINC MODE WITH THE AC CLEAR.
  423   000441  0017          COM                             /COMPLEMENT THE ACCUMULATOR.
  424   000442  0002          PDP                             /COME BACK TO PDP-8 MODE.
  425   
  426                 /       ON A PDP-8/I, THE "LINC" INSTRUCTION IS ONLY MEANINGFUL IF A DEVICE 14
  427                 /       INTERFACE IS PRESENT.  IN GENERAL, THIS IS NOT THE CASE; HOWEVER, IF A DEVICE
  428                 /       14 INTERFACE IS PRESENT, THE ACCUMULATOR MIGHT CHANGE AND/OR THE NEXT
  429                 /       OPERATION MIGHT BE SKIPPED.  THE "COM" AND "PDP" INSTRUCTIONS ON MACHINES
  430                 /       OTHER THAN THE PDP-12 ARE HARMLESS "AND 0017" AND "AND 0002" INSTRUCTIONS
  431                 /       RESPECTIVELY.  AS SUCH, IT IS IMPOSSIBLE FOR THE ACCUMULATOR TO BE SET TO 7777
  432                 /       AS LONG AS LOCATION 0002 IS NOT SET TO 7777; THE DETECTION ROUTINE SHOULD
  433                 /       ENSURE THIS IS NOT THE CASE TO PREVENT POTENTIAL CONFUSION.
  434   
  435   000443  7001          IAC                             /ADD ONE [MAKING THE AC 0000 ONLY ON A PDP-12].
  436   000444  7650          SNA CLA                         /SKIP IF NOT RUNNING ON A PDP-12.
  437   000445  5273          JMP     PDP12                   /RETURN WITH AC SET FOR PDP-12 DETECTION VALUE.

  439                 /       FOR CONSISTENCY, NO PERIPHERAL CONSIDERATIONS ARE USED TO DISTINGUISH MODELS.
  440                 /       AS SUCH, JUST PERFORM EXIT FOR PDP-8/I.
  441   
  442   000446  5275          JMP     PDP8I                   /RETURN WITH AC SET TO PDP-8/I DETECTION VALUE.
  443   
  444                 /       COMES HERE TO DETERMINE WHICH OMNIBUS PROCESSOR [OR NEWER] IS IN EFFECT.
  445   
  446                 /       ON THE KK8F PROCESSOR [FOUND ONLY IN PDP/8E, PDP-8/F, PDP-8/M AND PDP-8/A-600
  447                 /       SERIES SYSTEMS], A SYSTEM-UNIQUE OPERATION IS AVAILABLE: EXECUTING A 7016
  448                 /       INSTRUCTION [WHICH IS THE SAME INSTRUCTION CODE AS ROTATE TWICE IN BOTH
  449                 /       DIRECTIONS] WILL LOAD THE ACCUMULATOR BITS[0-4] WITH THE CURRENT PAGE OF THE
  450                 /       INSTRUCTION; BITS[5-11] OF THE ACCUMULATOR ARE LOADED FROM THE EXECUTED
  451                 /       INSTRUCTION.  FOR EXAMPLE, IF THE INSTRUCTION IS EXECUTED ANYWHERE ON PAGE 1
  452                 /       [0200 THROUGH 0377], THE ACCUMULATOR IS SET TO 0216.
  453   
  454                 /       ON THE KK8A PROCESSOR [FOUND ONLY IN PDP-8/A-400 SERIES AND SIMILAR SYSTEMS],
  455                 /       EXECUTION OF 7016 WILL LOAD THE ACCUMULATOR WITH THE UPDATED PROGRAM COUNTER.
  456                 /       FOR EXAMPLE, IF THE INSTRUCTION IS EXECUTED IN LOCATION 0215, THE ACCUMULATOR
  457                 /       IS SET TO 0216.
  458   
  459                 /       PRACTICAL DEVICE HANDLERS HAVE BEEN WRITTEN FOR OMNIBUS-BASED SYSTEMS THAT
  460                 /       EXPLOIT THIS FEATURE BY PLACING THE 7016 INSTRUCTION AT RELATIVE LOCATION 015
  461                 /       ON THE CURRENT PAGE; THE ACCUMULATOR WILL BE CHANGED TO THE SAME VALUE
  462                 /       REGARDLESS OF WHICH OF THE TWO POSSIBLE PROCESSORS IS PRESENT.  THIS ALLOWS
  463                 /       RELOCATABLE HANDLERS TO BE WRITTEN USING THE SHORTEST POSSIBLE CODE.
  464   
  465   000447  7016  PDPOMNI,RTL!RTR                         /EXECUTE INSTRUCTION WHICH CHANGES THE
  466                                                         /ACCUMULATOR ON PDP-8/E AND PDP-8/A SYSTEMS.
  467   000450  7450  UPDTPC, SNA                             /SKIP IF NOT A NEWER MODEL [WHICH DOES NOT
  468                                                         /CHANGE THE ACCUMULATOR].
  469   000451  5261          JMP     PDPCHIP                 /JUMP IF NEWER CHIP-BASED PROCESSOR.
  470   
  471                 /       WE NEED TO DETERMINE WHICH CPU [KK8F OR KK8A] IS IN EFFECT.
  472   
  473   000452  1373          TAD     (.&7600+16^7777)        /CHECK FOR PDP-8/E VALUE.
  474   000453  7450          SNA                             /SKIP IF NOT PDP-8/E.
  475   000454  5272          JMP     PDP8E                   /SET PDP-8/E RETURN VALUE AND EXIT.
  476   
  477                 /       ON THE KK8A, THE AC SHOULD BE THE UPDATED PROGRAM COUNTER WHERE THE 7016 WAS
  478                 /       EXECUTED.  NOTE: IT IS CONCEIVABLE THIS IS A NON-DEC "HOME-BREW" CPU BASED ON
  479                 /       OTHER HARDWARE.
  480   
  481   000455  1372          TAD     (.&7600+16-UPDTPC)      /CHECK FOR PDP-8/A VALUE.
  482   000456  7650          SNA CLA                         /SKIP IF SOME OTHER CPU.
  483   000457  5271          JMP     PDP8A                   /SET PDP-8/A RETURN VALUE AND EXIT.
  484   
  485                 /       THE CPU IS NOT A KNOWN DEC MODEL.  SEVERAL PEOPLE HAVE CREATED THEIR OWN CPU
  486                 /       USING ALTERNATE TECHNOLOGY.
  487   
  488   000460  5302          JMP     CPUNKNOWN               /SET RETURN VALUE FOR UNKNOWN SYSTEM AND EXIT.

  490                 /       COMES HERE TO DETERMINE WHICH OF THE PREVAILING PDP-8 CHIP-BASED PROCESSORS IS
  491                 /       IN EFFECT.  THE IM/HM6100 CHIP IS ESSENTIALLY FULLY COMPATIBLE WITH THE KK8E
  492                 /       AND KK8A PROCESSORS [WITHOUT THE IMPLEMENTATION OF THE 7016 OPERATION OF
  493                 /       THE OTHER TWO PROCESSORS].
  494   
  495                 /       IF FURTHER TESTING WERE TO BECOME NECESSARY, THERE IS AN ADDITIONAL
  496                 /       COMPATIBILITY QUIRK THAT CAN BE EXPLOITED: FOR ALL OTHER 12-BIT MODELS, WHEN
  497                 /       AN INSTRUCTION IS SETUP TO USE AN AUTO-INDEX REGISTER LOCATION, THE OPERATION
  498                 /       ALWAYS OCCURS AS EXPECTED.  THIS IS TRUE EVEN IF THE INSTRUCTION IS EXECUTED
  499                 /       ON PAGE ZERO.  ADDITIONALLY, IF THE INSTRUCTION IS MODIFIED TO SET THE CURRENT
  500                 /       PAGE BIT, THE AUTO-INDEXING OPERATION STILL OCCURS BECAUSE THE EFFECTIVE
  501                 /       POINTER ADDRESS IS IN THE RANGE OF 0010 THROUGH 0017.  HOWEVER, ON THE
  502                 /       IM/HM6100 CHIP, THE AUTO-INDEXING DOES NOT OCCUR IF THE CURRENT PAGE BIT IS
  503                 /       SET WHEN THE INSTRUCTION IS EXECUTED ON PAGE ZERO.  SINCE. AS A PRACTICAL
  504                 /       MATTER, THIS PARTICULAR PROGRAMMING METHOD IS NEVER USED, THIS QUIRK IS
  505                 /       GENERALLY UNIMPORTANT.  ADDITIONALLY, SUCH AN INSTRUCTION CANNOT BE DIRECTLY
  506                 /       CREATED BY ANY OF THE PREVAILING ASSEMBLERS, ALL OF WHICH WILL FORCE THE
  507                 /       CURRENT PAGE BIT OFF USING CONVENTIONAL PROGRAMMING TECHNIQUES.  THE PROBLEM
  508                 /       WAS DISCOVERED BY EXECUTING CERTAIN PDP-8/E-ORIENTED DIAGNOSTIC PROGRAMS THAT
  509                 /       ACTUALLY CHECK FOR THIS SET OF CIRCUMSTANCES.  THE INCOMPATIBILITY HAS BEEN
  510                 /       NOTED ADVISING TO AVOID RUNNING THESE DIAGNOSTIC PROGRAMS ON AFFECTED SYSTEMS
  511                 /       SUCH AS THE VT78 AND THE INTERSIL INTERCEPT I.
  512   
  513                 /       THE QUIRK OF THE IM/HM6100 CHIP WAS CORRECTED IN THE HM6120 CHIP, WHICH WOULD
  514                 /       MAKE CPU DETECTION SLIGHTLY MORE DIFFICULT; HOWEVER, THE HM6120 ADDS A NEW
  515                 /       UNIQUE FEATURE: A NEW GROUP I OPERATE WAS ADDED TO ROTATE THE ACCUMULATOR
  516                 /       3 BITS TO THE LEFT [CIRCULARLY EXCLUDING THE LINK].  THIS FEATURE WILL BE USED
  517                 /       TO DIFFERENTIATE BETWEEN THE TWO CHIP-BASED IMPLEMENTATIONS.
  518   
  519   000461  7215  PDPCHIP,CLA IAC RAL!RAR                 /THIS WILL LOAD THE AC WITH 0010 ON THE HM6120.
  520   000462  1371          TAD     (-10)                   /COMPARE TO EXPECTED VALUE FOR HM6120.
  521   000463  7650          SNA CLA                         /SKIP IF NOT A MATCH.
  522   000464  5267          JMP     CP6120                  /SET 6120 RETURN VALUE AND EXIT.
  523   
  524                 /       SINCE THERE ARE APPARENTLY NO OTHER MODELS TO CHECK FOR, THE EXPANDED TEST FOR
  525                 /       THE QUIRKS OF THE IM/HM6100 WILL NOT BE REQUIRED.
  526   
  527   000465  5270          JMP     CP6100                  /SET 6100 RETURN VALUE AND EXIT.
  528   
  529   000466  0000  MCHTEMP,.-.                             /LOCATION 0000 SAVED HERE DURING PDP-5 TEST.

  531                 /       RETURN GAUNTLET.
  532   
  533                 /       THE SUBROUTINE EXITS WITH THE ACCUMULATOR SET TO THE VALUE CORRESPONDING TO
  534                 /       THE CPU MODEL DETECTED.
  535   
  536   000467  7001  CP6120, IAC                             /SET 6120 VALUE.
  537   000470  7001  CP6100, IAC                             /SET 6100 VALUE.
  538   000471  7001  PDP8A,  IAC                             /SET PDP-8/A VALUE.
  539   000472  7001  PDP8E,  IAC                             /SET PDP-8/E VALUE.
  540   000473  7001  PDP12,  IAC                             /SET PDP-12 VALUE.
  541   000474  7001  PDP8L,  IAC                             /SET PDP-8/L VALUE.
  542   000475  7001  PDP8I,  IAC                             /SET PDP-8/I VALUE.
  543   000476  7001  LINC8,  IAC                             /SET LINC-8 VALUE.
  544   000477  7001  PDP8S,  IAC                             /SET PDP-8/S VALUE.
  545   000500  7001  PDP8,   IAC                             /SET PDP-8 VALUE.
  546   000501  7001  PDP5,   IAC                             /SET PDP-5 VALUE.
  547   000502  5600  CPUNKNO,JMP I   MACHINE                 /RETURN TO CALLER; AC IS CLEAR
  548                                                         /FOR UNKNOWN MODEL TYPE.
  549   
  550          *0571 
        000571  7770 
        000572  7746 
        000573  7362 
        000574  7700 
        000575  0002 
        000576  4002 
        000577  0500 
               *0600          PAGE

  552   000600  2410  OPNMSG, TEXT    "THIS PROGRAM IS RUNNING ON A"
        000601  1123 
        000602  4020 
        000603  2217 
        000604  0722 
        000605  0115 
        000606  4011 
        000607  2340 
        000610  2225 
        000611  1616 
        000612  1116 
        000613  0740 
        000614  1716 
        000615  4001 
        000616  0000 
  553   
  554                 /       MODEL-SPECIFIC TEXT STRINGS.
  555   
  556   000617  1640  UNKNMSG,TEXT    "N UNKNOWN MODEL."
        000620  2516 
        000621  1316 
        000622  1727 
        000623  1640 
        000624  1517 
        000625  0405 
        000626  1456 
        000627  0000 
  557   000630  4020  PDP5MSG,TEXT    " PDP-5."
        000631  0420 
        000632  5565 
        000633  5600 
  558   000634  4020  PDP8MSG,TEXT    " PDP-8."
        000635  0420 
        000636  5570 
        000637  5600 
  559   000640  4020  PDP8SMS,TEXT    " PDP-8/S."
        000641  0420 
        000642  5570 
        000643  5723 
        000644  5600 
  560   000645  4014  LINC8MS,TEXT    " LINC-8."
        000646  1116 
        000647  0355 
        000650  7056 
        000651  0000 
  561   000652  4020  PDP8IMS,TEXT    " PDP-8/I."
        000653  0420 
        000654  5570 
        000655  5711 
        000656  5600 
  562   000657  4020  PDP8LMS,TEXT    " PDP-8/L."
        000660  0420 
        000661  5570 
        000662  5714 
        000663  5600 

  564   000664  4020  PDP12MS,TEXT    " PDP-12."
        000665  0420 
        000666  5561 
        000667  6256 
        000670  0000 
  565   000671  4020  KK8FMSG,TEXT    " PDP-8/E OR SIMILAR MODEL."
        000672  0420 
        000673  5570 
        000674  5705 
        000675  4017 
        000676  2240 
        000677  2311 
        000700  1511 
        000701  1401 
        000702  2240 
        000703  1517 
        000704  0405 
        000705  1456 
        000706  0000 
  566   000707  4020  KK8AMSG,TEXT    " PDP-8/A."
        000710  0420 
        000711  5570 
        000712  5701 
        000713  5600 
  567   000714  4066  CHP6100,TEXT    " 6100 CHIP-BASED MACHINE."
        000715  6160 
        000716  6040 
        000717  0310 
        000720  1120 
        000721  5502 
        000722  0123 
        000723  0504 
        000724  4015 
        000725  0103 
        000726  1011 
        000727  1605 
        000730  5600 
  568   000731  4066  CHP6120,TEXT    " 6120 CHIP-BASED MACHINE."
        000732  6162 
        000733  6040 
        000734  0310 
        000735  1120 
        000736  5502 
        000737  0123 
        000740  0504 
        000741  4015 
        000742  0103 
        000743  1011 
        000744  1605 
        000745  5600 
  569   000746  4004  DMATMSG,TEXT    " DECMATE."
        000747  0503 
        000750  1501 
        000751  2405 
        000752  5600 

57 57
58 58

  571                         $                               /THAT’S ALL FOLK!

34 34

36 36
37 37

ASSEMBLY STATISTICS

NO ERRORS DETECTED 

NO LINKS GENERATED

5K MEMORY UTILIZED 

1 FILE  CREATED

54 SYMBOLS

1 1
2 2
3 3
4 4
5 5
6 6
7 7
8 8
9 9

10 10
11 11
12 12
13 13
14 14
15 15
16 16
17 17
18 18
19 19
20 20
21 21
22 22
23 23
24 24
25 25
26 26
27 27
28 28
29 29
30 30
31 31
32 32
33 33
34 34
35 35
36 36
37 37
38 38
39 39
40 40
41 41

44 44
45 45
46 46
47 47
48 48
49 49
50 50
51 51
52 52
53 53
54 54
55 55
56 56

NO MAPPING RESTRICTIONS

00000 00100000 10000000 00000000 00000000 00000000 00000000 00000000 00000000 
00100

00200 11111111 11111111 11111111 11111111 11111111 11111111 11111111 11111111 
00300 11111111 11111111 11110000 00000000 00000000 00000000 00000011 11111111 

00400 11111111 11111111 11111111 11111111 11111111 11111111 11111111 11111111 
00500 11100000 00000000 00000000 00000000 00000000 00000000 00000000 01111111 

00600 11111111 11111111 11111111 11111111 11111111 11111111 11111111 11111111 
00700 11111111 11111111 11111111 11111111 11111111 11100000 00000000 00000000 

01000
01100

01200
01300

01400
01500

01600
01700

02000
02100

02200
02300

02400
02500

02600
02700

03000
03100

03200
03300

03400
03500

03600
03700

32 32

04000
04100

04200
04300

04400
04500

04600
04700

05000
05100

05200
05300

05400
05500

05600
05700

06000
06100

06200
06300

06400
06500

06600
06700

07000
07100

07200
07300

07400
07500

07600
07700


